FDO & VPD

(Teaching a blind octopus how to juggle)

Introduction

FDO is a programming language used by AOL developers to create dynamic forms and  objects in conjunction with other AOL technologies.
It is the primary language behind all form objects on the AOL service world wide.

Used in conjunction with Visual Publisher Designer, FDO can dramatically improve the functionality of basic forms. This document is intended to give AOL developers a basic understanding of some of the more common uses of FDO with VPD. It is not intended to be a course in programming, nor should it be seen as an in-depth guide to atom strings more commonly used in client development or localisation.

If you read and understand the contents of this document, you should be armed with a selection of FDO tricks and tips that you can use from day to day and hopefully enough understanding of the programming syntax to be able to develop your own tricks.

I may in later chapters include information specific to creating forms using FDO within the form_edit environment but I’m not making any promises.

Tools you will need

This document makes continual reference to the FDO 91 manual which lists and details the basic syntax of all the atom strings we will be using. If you do not have a copy of this manual, you should be able to find it at Keyword: TECHDOCS.

As this manual revolves  around the use of FDO with the Visual Publisher Designer environment, it is assumed that you will have at least a moderate knowledge of this tool and its features.

No FDO developers toolkit  is complete without the wonderful Atom Debug tool for the AOL client. Details of where you can download this tool can be obtained from you local production support department.

Chapter 1 

Your first peek at FDO

FDO is the language behind AOL form objects as created in VPD. Before the introduction of VPD, all forms were created on a UNIX based system called ‘Form Edit’. It is still possible to use Form Edit to create and manipulate forms. Indeed there are many atom strings which can only be used in this environment. However for now, we’ll just mention it in passing.

Right, let’s go meet the octopus. *
We’re going to take a closer look at how VPD uses FDO to create a form.

1.  Sign on to AOL, activate host access on your area manager menu and create a new blank form.

2.  Next, go to ‘Form Operations’ on your Designer menu and select ‘Toggle Visual Mode’.

You will see a form with some FDO atom strings like the one shown below. 

Figure 1
These atom strings define what a blank form looks like. Although the form looks blank and there are no objects in the tree view, we can already see that even a simple white form has behind it quite a lot of commands that define the size and shape of a blank form. 

The atom strings that make up a blank form are quite boring, so for now, just have a look around and see if you can tell what each atom string does. Its a good idea to go through the FDO manual and translate each string individually into English, but not now. (We’ll come back to this when we talk about creating non-visual forms such as dispatches).

Go back to the Designer menu and toggle visual mode again to see the blank form. Next, drop a button from the object bar on to the form. We are going to use this to activate our very first atom string.

Double click on the button to display the properties box, and select the action tab. You have probably already used this to set the action of the tab using the drop-down list.

Leave the event as ‘Selection’ and for the action, select FDO from the second drop down menu. A blank box appears below. This is where we will be typing in our raw FDO. 

Right, here goes….

Type the following atom string into the box:

uni_start_stream

async_alert <warning, ”Hello World!”>

uni_end_stream
Next, click on the ‘Compile’ button above where you have typed your code. If you have typed in the atom stream correctly, nothing will happen, otherwise, your typos will be highlighted. 

Always compile your atom streams before moving on. If you are writing a particularly long stream, compile regularly to avoid going back through your code to correct any errors you have cut and pasted in. (I use a customised MSWord dictionary to write my atom streams in Word and let the spell checker sort it out.)

Now, preview the form (CTRL ‘D’) and click on the button. You should see this:

Figure 2
Now lets talk about what we’ve just done. First of all we placed our FDO on the action tab of the button. We did this because we have to have some way of initialising the FDO stream. There are many different ways of doing this, some of which you are probably already familiar with, here’s just a few.

1.  Using the action tab on a button or similar form object.

2.  Using one of the predetermined events listed in the event list of the form. (eg ‘Create’.)

3.  Using the FDO in-stream or post-stream of the form or of an object on a form. 
I will explain later when and where is the most appropriate place to put your atom strings, for now the important thing to remember is that:

 FDO strings don’t just happen. They have to be activated in some way. 
That’s the first basic principal of FDO programming out of the way, now lets look at our atom string again.

uni_start_stream

async_alert <warning, ”Hello World!”>

uni_end_stream

VPD has an implied “act_replace_select_action” before the uni_start_stream you put in the beginning of your stream.  It’s important to remember that if you are ever using form_edit to create forms and then flipping back and forth between that and VPD. If you leave this atom out when building forms in form_edit, your button actions will be activated upon rendering of the form and not when the buttons are clicked.  



The first line as always is important. This defines the start of an atom string. Similarly, the third line defines the end of and atom string. Usually when you are writing atom strings within VPD, you can leave this and the last line out, as VPD automatically inserts these lines in the relevant section of the form full stream. However its good practice to always include these lines at the beginning and ending of any new atom string. This becomes even more important when you start working with multiple events and sub routines.

As you can see, these atoms are divided into three sections, separated by an underscore. The first section identifies which protocol the line is invoking.

Phew! That was our first bit of jargon. Time for the analogy.

You might like to think of protocols as the various body parts of our octopus. Before we tell the octopus to pick up a ball, we have to first tell the octopus to use its arm.  An atom string with the prefix uni_ invokes the universal protocol. In terms of our octopus, I compare the universal protocol to the brain of the octopus. So a rough translation of the atom string uni_start_stream might be 

‘OK, listen to this and do what I say.’

Similarly, the last line might be 

‘OK you can stop listening to me now.’

If you look through the FDO manual, you will notice that each chapter covers a different protocol and describes the functions of the atom strings contained within that protocol. Each protocol has a defined set of atom strings that you can use and its important to understand that you cannot swap and change commands from one protocol to another. (For example, the octopus can’t use it’s eyes to move something, or remember something with its legs).

We’ll talk a bit about some of the other protocols in the next section.  For now, we’ll look at the last remaining atom in our FDO string.

aysnc_alert <warning, ”Hello World!”>

It doesn’t take much imagination to figure out what this line does. 

This atom belongs to the ASYNC protocol. (Asynchronous protocol) The ASYNC protocol mainly handles communications between the AOL Host and the client. It also has a few miscellaneous functions such as displaying an alert dialog box.

The atom ‘async_alert’ tells the octopus to display a small grey dialog box. The details following this atom in brackets are know and ‘arguments’. It is these arguments that tell the octopus what type of dialog box to display and what text to display in it. The arguments for the async_alert atom must be specified as follows:

async_alert <type of dialog box , “text to be displayed”>

You can find out more about what different types of argument can be specified with async_alert on page 10-2 of the FDO manual.
Many of the atoms you will be using must have arguments specified between brackets like this.  Some atoms will not work if there is no argument specified, others take a default argument unless another is specified. We’ll talk a bit more about arguments when we come across other atoms that use them.

It is possible to define the alert box from within the atom stream by describing its size and colour and so on, but that would be making things more complicated than they need to be and trust me, you don’t want that. Things will get complicated enough pretty soon.

Next we’re going to talk about some of the other protocols and atoms you can use with them and show you how to so your first bit of FDO trickery by making content disappear and reappear dynamically. (AKA Hide and Reveal).

Chapter 2 

Hide and Reveal

So far you have learnt how to format a simple atom stream as the action of a button. In this chapter, we are going to write a new atom stream that will manipulate the contents of a form. In the process you will learn some more atom strings that can be used to dynamically manipulate the contents of a form.

First of all we are going to need to give our octopus some balls to juggle with. Go to the object menu and drop a label onto the form. Click on the label to bring up the properties box and insert the following text into the content tab. “What’s your favorite colour?”.

Next we are going to need two buttons on the form. You can use the button we placed on the form in chapter 1 or delete it and add two fresh buttons if you prefer.

Now we will need two more labels, one with the contents “My favorite colour is Blue” and the other with the contents “My favorite colour is Red”.

Place these objects anywhere on the form.

Now in a few minutes we are going to write a set of instructions that will tell the octopus to hide or reveal the appropriate label , depending on which button is clicked. In order to do this, we will need to identify each object individually so that the octopus can tell which object we are talking about. So we will need to click on the ID tab of each object’s property box and give it a unique ID.

Again, this should be nothing new to you. Now you should end up with a form that looks something like this. 

Figure 3
Note: in order for the following atom stream to work as it is written here, your object Ids should be the same as in the above diagram.

Now click on the visibility tab of each of the Red and Blue labels and uncheck the Visible check box. (Now if you preview the form, you should only be able to see the first label and the two buttons.)

Next, pick one of the buttons and change it’s title to “Blue”. Then click on the action tab, and choose FDO from the action drop down list.

Insert the following atom string into the box for the Blue button.

uni_start_stream

man_set_context_relative <5>

mat_bool_hidden <no>

mat_bool_invisible <no>

man_change_context_relative <4>

mat_bool_hidden <yes>

mat_bool_invisible <yes>

man_end_context

man_update_display

uni_end_stream

Next you will need to repeat this process for the ‘Red’ button only this time use the following atom string:

uni_start_stream

man_set_context_relative <5>

mat_bool_hidden <yes>

mat_bool_invisible <yes>

man_change_context_relative <4>

mat_bool_hidden <no>

mat_bool_invisible <no>

man_end_context

man_update_display

uni_end_stream

Now preview your form and see what happens when you click on the buttons. The appropriate label is displayed when you click on the red or blue buttons.

Lets have a close look at the FDO. You will have noticed that both buttons have the same atom stream apart from one or two values.

The first line we have already seen. The second line is brand new.

man_set_context_relative <4>

This atom invokes the Display Manager protocol. All atoms that fall within this protocol start with the prefix ‘man_ ‘.

In terms of our octopus, the display manager protocol serves as the eyes. It controls what the Octopus looks at and what it chooses to ignore.

Man_set_context_relative <4> Tells the octopus to look at the object with the relative ID  ‘4’*. Until we tell it to look elsewhere the octopus will apply any subsequent atoms to this object. 

In FDO programming terms, this line sets the object with the ID of ‘4’ as the Current Object* and/or context .

*The concept of the ‘current object’ is important in FDO programming. Although our octopus has many arms, he is only capable of doing one thing at a time. If we try and manipulate another object while his attention is still on the ‘current object’ it will become confused and in some cases, he will crash your client. In times to come, as you sit and stare at the monitor trying to figure out why your atom streams don’t work, you will find that tracing the path of the current object is often a good place to start. 

While we are on the subject, lets skip down a few lines and talk about:

man_end_context

This line tells the octopus to reset the current object to what was previously in context.
The default for the current object is usually the form object itself. So any atoms that follow this line will be applied to the form overall until the context is reset by another ‘man_set_context_relative’.

Another point to remember here is that our octopus is easily confused. When you are creating long atom streams that involve a variety of switching from one current object to another, it is always wise to be careful to use ‘man_end_context’ to reset the current object to avoid our poor little octopus getting lost within the object level of a form.

There are two more display manager protocols in this atom stream. The first is:

man_change_context_relative <4>

This atom allows us to change the current object from <4> to <5> without ending the context and resetting it. It’s a shortcut atom that replaces two lines of code with one.

The final display manager atom is:

man_update_display 

This is a very important atom. It tells the octopus to refresh the screen. If you do not include this atom, the changes will be made but will not show up on the screen. Its good practice to place a ‘man_update_display’ after any subroutine that alters the appearance of the form, and not just at the end of a long atom stream.

The only other atoms remaining are:

mat_bool_hidden <yes> / <no>

mat_bool_invisible <yes> / <no>
Both of these atoms belong to the Attribute Manager Protocol. The attribute manager protocol deals with the attributes or characteristics of objects within FDO. Any forms that define the appearance of an object, it’s height, size, colour etc, all belong to this protocol.

mat_bool_hidden <yes / No> 
Sets the current object as either hidden or not hidden. This atom is more important when we are dealing with forms that have relative positioning. When the form is being created, the code will adjust the size of the form depending on the objects on the form that are not hidden.

The second atom, acts in a similar way.

mat_bool_invisible <yes> / <no>

This sets the visibility of the current object to either visible or invisible.

So now I bet you’re asking – “Why use both?” Well, technically, you should be able to use just one of these strings to get the desired result. In my experience, you need to use both atoms together to make sure it always works. Best practice is to always use them together. 

(Some developers disagree on this point. But I always do this as a matter of default and I’ve not encountered any problems doing this. You can make your own mind up about how to do this but I strongly recommend this method, at least until you’re more confident with FDO)

Now lets have one last look at the entire atom stream: 

Atom




Translation
uni_start_stream



Hello? Listen up!
man_set_context_relative <5>

Look at object <5>
mat_bool_hidden <yes>


Hide it

mat_bool_invisible <yes>

Make it invisible

man_change_context_relative <4>
Now look at object <4>

mat_bool_hidden <no>


Reveal it

mat_bool_invisible <no>


Make it visible
man_end_context



Look at the form overall
man_update_display


Redraw the form
uni_end_stream



Ok stop listening
Right then, you’ve just created your first Click Trick in FDO. Save the form to your area manager, next we’re going to use it again in the next chapter to make it a bit more sophisticated.

Chapter 3 Basic Data Extraction

So now you know how to hide and reveal objects with FDO, buts its not very slick is it? You only have two options, red or blue. What if your favorite colour is ‘Champagne Gold’? 

The next step is to replace the buttons with an input box and alter the content of the form dynamically. 

First off, lets delete the Red and Blue buttons and one of the text fields from our form.

Then add an input box and one new button. So that your form looks like this:

Figure 4
Now, the next bit of FDO code is a bit of a jump. We place this on the action tab of the button, with the ‘Selection’ event

uni_start_stream

man_set_context_relative <1>

de_start_extraction <0>

de_set_data_type <text>

de_get_data_pointer

uni_save_result

de_end_extraction

man_change_context_relative <2>

uni_get_result

uni_use_last_atom_string <man_replace_data> 
man_end_context

man_update_display

uni_end_stream
Right, straight away you’ve probably noticed a brand new Protocol. ‘de_’ this is the Data Extraction protocol, the arm that the octopus uses to take information form a form.

All data extraction subroutines must follow the same format. First of all, they should all begin with;

de_start_extraction <0>

This atom initiates the data extraction protocol. 

The <0> is a flag that can be set from 0-6 depending on the intended destination of the data. The value <0> tells the octopus that the intended destination is local. (ie: another object or variable on the same form.) You shouldn’t need to worry about the other values for this atom. When working in VPD, you’ll be using the <0> ninety-five percent of the time.

Similarly, the end of the sub routine is marked with:

de_end_extraction

The next atom is:

de_set_data_type <text>
This atom tells the octopus what kind of data you want it to extract. There are three main data types. Text (ASCII characters), Values (Numbers) and Raw Data (Hex Code).

The atom string will probably work if we don’t specify a data type. Sometimes it will, sometimes it won’t. Best bet is to always specify a data type, that way when your extraction atoms don’t work, you know that your problem lies elsewhere.

GFP! – GPF! – GPF! –GPF! – GPF! – GPF! – GPF! 

The data extraction protocol is one of the more unstable protocols and errors in data extraction subroutines can often result in GPF’s. Bear this in mind before you start trying shortcuts.

de_get_data_pointer

This atom does the work. It gets the information form the current object and returns a pointer to

the current register as a String.
 What does this mean? 

We’ll talk a bit more about pointers and ‘the current register’ when we come to variables in the next section, basically this means that the atom will send the actual text to the host rather than the value (Hex Code) of the data.

As well as the current object, FDO also uses a principal called the Current Register. The current register is an area of the octopus’s brain that it uses to ‘think’ about the current object. The current register at any point in time refers to the bit of the octopus’s brain that it is using to temporarily register any information about the current object. The current register is reset every time the context changes or is reset. So we will need to move this returned value to another register if we want to refer to it later, after we have reset the current object to a different object on the form. 

Each atom that is executed returns a value to the current register. Most of the time this returned value is either true or false (1/0). But some atoms, like ‘de_get_data_pointer’ atom, return a specific value according to the preset data type. (In this case text).

The next atom:

uni_save_result
Makes sure that the returned value is remembered. The octopus, saves the returned pointer in it’s short term memory. (The prefix ‘uni_’ makes sure that the returned value is saved at the Global level, so it can be retrieved after the context is set to another object or form. You could set context to the root or the toolbar or another form and still be able to use uni_get_result to retrieve the string.)

In the next chapter, you will learn how to use different atoms to save this data with its own ID as a variable, but for now, we are just going to leave it there as a piece of data on the form level that has no unique identity.

We need to do this because we are going to change the current object in a minute and unless we tell the octopus to remember the data, he will forget it as soon as his attention is redirected to another object.

The next atom: ‘de_end_extraction’ simply marks the end of the data extraction subroutine. Your FDO might work if you leave this out, but usually not so you can consider this as a requirement when working with the data extraction protocol.

 So lets just recap and see where we are in the FDO.

At this point in the atom stream we have extracted the data form the input box and saved it at the form level for future reference. Next we need to take this data and paste onto the end of the text in field <2>

Here’s how we do this:

man_change_context_relative <2>

This changes the current object to field <2>.

uni_get_result

This calls the data that we have already saved back into the octopus’s short term memory. If we had not saved the data earlier on, the FDO would not be able to find the ‘result’ because it would be looking for it within the current object instead of at the form level.

uni_use_last_atom_string <man_append_data>

This atom is a combination of two different atoms. The first ‘uni_use_last_atom_string’ uses the last returned value (the ‘_string’ identifies it as text) to execute the atom in brackets. So in this case, our ‘uni_get_result’ atom will return the text we saved earlier as the last returned value. 

The ‘man_append_data’ atom, is used to append data onto the end of the data contained in the current object. Effectively this is the same as executing the atom :

‘man_append_data <”XXX”>’

(where XXX equals the contents of the input field <1>)
The last three atoms reset the current object back to the form level, update the display and end the stream.

So here’s what your form should look like now when you preview it:

Figure 5
Now save this form again. If you play around with this form you will probably notice some shortfalls. For example, what happens when you click on the button without anything in the input box?

In the next chapter we are going to concentrate on working with variables and how you can use them to build in error-checking subroutines to give your forms a bit more functionality. 

Chapter 4 

Working with Variables in FDO

We’ve already talked about the Current Register in the previous chapter. Now we’re going to go into a little more detail.

When the context has been set to an object on a form, any returned values are returned to the current register, which is re set when the context changes. This means that any data we wish to keep or manipulate at a later stage, we need to save at the form level. Otherwise the data will be lost when the current object changes and the current register is re set. In our last example we used the atom ‘uni_save_result’ to store the data at the form level and called it back to the current register by using the atom ‘uni_get_result’.

In effect we have already used a variable, as the returned value reflects whatever data the user typed in the input box. However, this approach would not work if we needed to store more than one piece of data.

In order to do this we need to load the returned values into variables with unique Ids. 

The current register can contain up to four different variables, called <A>, <B>, <C>, and <D>. Each of these variables can hold three different types of data. Text (String), a value (Num) and raw data (Data). In effect this means that you can have up to twelve different pieces of data stored in variables at any one time, although outside of client forms, its hard imagine a case where you would need to manipulate that much text, numbers and raw data in one go. 

Here’s where the juggling comes in…

Once you have your data stored in variables, its possible to manipulate them by adding them together, subtracting them, sending them to the host and so on. However, there is one main disadvantage to using FDO to manipulate variables and its important to know it before we get all excited.

Most atoms that deal with variable manipulation work with the  A and B registers only. This means that you will have to copy the content from your C or D variables into the A or B registers before you can get anything done. This is what I mean about ‘teaching the octopus how to juggle’.

‘Buckets’

When dealing with numbers, it is possible to store more than four variables by giving each variable a further id e.g. “<A, 45>”. But this is tricky and fraught with pitfalls. For example, when we store variables like this, they are stored at the object level and so won’t be available to call if context changes to another object. The method of storing variables inside objects is commonly referred to amongst the FDO community as storing variables in ‘Buckets’. 

We’ll go into storing data as variables in buckets later on, but for now, lets just concentrate on the basics.

Lets have a look a quick look at the current register.

The Current Register

A
B
C
D

Num
String
Data
Num
String
Data
Num
String
Data
Num
String
Data

Right, that’s the basics out of the way. Now lets go and include variables into our form.

Figure 6
First of all we are going to save the input as a string variable. Here’s what the atom stream should look like:

uni_start_stream

man_set_context_relative <1>

de_start_extraction <0>

de_set_data_type <text>

de_get_data_pointer

var_string_set_from_atom <A>

de_end_extraction

man_change_context_relative <2>

var_string_get <A>

uni_use_last_atom_string <man_append_data>

man_end_context

man_update_display

uni_end_stream

If you compare this stream to the previous stream, you will notice that we have simply replaced the atoms ‘uni_save_result’ and ‘uni_get_result’ with the atoms ‘var_string_set_from_atom <A>’ and ‘var_string_get <A>’.

The first atom ‘var_string_set_from_atom <A>’ loads whatever was returned to the current register by the last atom into a variable with the Id <A>.

The second atom ‘var_string_get <A>’ returns the contents of the variable <A> into the current register.

This works exactly the same way as the previous atom stream, the only difference is that instead of the input being stored in the current register, it is stored as a variable, leaving us free to manipulate that data or load another variable with different data from the user.

Straight away you will notice that we have invoked a brand new protocol, the ‘var_’ protocol which deals with the creation and manipulation of variables. Chapter XX of the FDO manual deals with the variable manager protocol. If you look through the chapter at the different atoms, you will see the kind of things we can now do with the variable which were not possible when it was saved using ‘uni_save_result’.

The advantage of this approach means that we can build in some simple error-checking.

Error Checking:

Its always important that you build error checking into your atom streams. It makes for a more complete form and minimises risks when working over different platforms and clients. It can also be handy when debugging complicated streams.

What we are going to do next is make sure that the input box contains content before we append the label at the end of the atom stream. We do this by examining the contents of the <A> variable and popping up an alert box to warn the user to enter some text into the input box before processing the rest of the atom stream.

Replace the FDO on the button with the following:

uni_start_stream

man_set_context_relative <1>

de_start_extraction <0>

de_set_data_type <text>

de_get_data_pointer

var_string_set_from_atom <A>

de_end_extraction

var_string_get <A>

if_stra_eq_null_then <1,2>

async_alert <warning, "Please enter a colour">
uni_end_stream

uni_sync_skip <1> 

man_change_context_relative <2>

var_string_get <A>

uni_use_last_atom_string <man_append_data>

man_end_context

man_update_display

uni_end_stream

uni_sync_skip <2>

Now when you click on the button with nothing in the input box this is what you should see this:

Figure 7
Now lets look at how this works.

var_string_get <A>

As we have already seen, this atom loads the contents of the string variable <A> into the current register.

if_stra_eq_null_then <1,2>

This is another new atom and another new protocol, the Conditional Protocol. Conditional Protocol atoms allow us to use simple If/Then statements to build Boolean logic into our atom streams. In this case the atom checks to see if the contents of string variable <A> are empty, and if so it executes all the atoms between this point and uni_sync_skip <1>.  If not, these atoms will be skipped and all the atoms between uni_sync_skip <1> and uni_sync_skip <2> will be executed instead.

async_alert <warning, "Please enter a colour">

uni_end_stream

We’ve seen both of these atoms in the first chapter.

uni_sync_skip <1>  and uni_sync_skip <2> 

These atoms act as markers that identify different subroutines within an atom stream. In this case we have used two markers, uni_sync_skip <1> and uni_sync_skip <2>. You can use only one argument for if you just need to do something extra if that condition is true.  For example… you could change that stream to this one:

uni_start_stream

man_set_context_relative <1>

de_start_extraction <0>

de_set_data_type <text>

de_get_data_pointer

var_string_set_from_atom <A>

de_end_extraction

var_string_get <A>

if_stra_eq_null_then <1>

async_alert <warning, "Please enter a colour">

uni_abort_stream (stops the stream
uni_sync_skip <1> 

man_change_context_relative <2>

var_string_get <A>

uni_use_last_atom_string <man_append_data>

man_end_context

man_update_display

uni_end_stream

;uni_sync_skip <2> (this wouldn’t be there

Note: When a conditional atom resolves as ‘True’ the last returned value will be ‘1’ and when it resolves as ‘False’ the last returned value will be ‘0’. You can use this to check the last returned value to build in error checking.

So now lets go through the whole atom stream one more time with English translations.

uni_start_stream

OK Listen up!

man_set_context_relative <1>

look at the input box

de_start_extraction <0>
I’m going to ask you to extract data form here in a minute

de_set_data_type <text>
The data will be an ASCII character string

de_get_data_pointer
Right, go get it 

var_string_set_from_atom <A>
Lets call that text <A> form now on

de_end_extraction
Ok that’s the extraction bit over with

var_string_get <A>
Look at the contents of the string bit of the <A> register.

if_stra_eq_null_then <1,2>
If its empty do the following, if its not carry on after uni_sync_skip <1> 

async_alert <warning, "Please enter a colour">
Display a warning to the user

uni_end_stream
Ok Stop.

uni_sync_skip <1>
Carry on from here 

man_change_context_relative <2>
Look at the label (and reset the current register)

var_string_get <A>
Look at the contents of the string bit of the <A> register.

uni_use_last_atom_string <man_append_data>
Append the contents of the label with the contents of the string variable <A>

man_end_context
Go back to the form level and reset the current register

man_update_display
Re-draw the form

uni_end_stream
Ok stop

uni_sync_skip <2>
Marks the end of the second subroutine

So that’s a basic introduction to the use of variables and conditional atoms in FDO. There’s lots more to learn here as you will see in the next few chapters. I recommend that you take the time to look over the relevant chapters in the FDO manual and familiarise yourself with the various atoms that fall within these two protocols.

To finish off we’re going to look at the atom streams we use to extract a search string typed by the user and launch a channel search for that search term.

****Click Trick******

Channel Search

The channel search works by passing a query string to a PERL script hosted on the AOL host. The query string looks and acts just like a URL. So we are going to us a buffer subroutine to pass this URL to the host.

Here’s the syntax of the URL /command line:

http://aolsearch.aol.co.uk/article.adp?query=XXXXrst=1&aol_last=10&test=0&channel=ADVHealth

Where XXXX = the words your searching for and ADVHealth = the name of the Rainman search database.

So what we have to do is construct that URL on the form, load it into a variable and pass it to the host by using a Buffer sub routine. Lets break it down step by step.

1. Extract the input form the user.

2. Append the input onto the first half of the URL.

3. Append the second half of the URL onto the end of that.

4. Submit the entire URL to the host.

Note: There’s a handy little atom ‘var_string_concatenate’ which on first glance would appear to make our life easier here. It takes two string variables and joins them together as one. However I have never been able to get this atom to work on the Mac platform. So until such time as the Mac client supports this atom, we are stuck using this method. 

First of all we need to make some changes to our form so that it looks like this:

Figure 8
You will notice that the relative Ids have changed slightly to fit in with standard Rainman mapping conventions for search on AOL UK. We’ve also added a new input box <271>. We are going to use this input box to contain the URL as we are constructing it. (When we are finished we will set the visibility of this object to invisible). Then we will extract the contents of this input box and pass them to the host.

Now put the following FDO onto the select action of the button:

uni_start_stream

  de_start_extraction <0>

  man_set_context_relative <270>

  de_get_data_pointer

  var_string_set_from_atom <A>

  de_end_extraction

  var_string_get <A>

  if_stra_neq_null_then <1, 2>

  man_change_context_relative <271>

  man_replace_data <"aol://1223:44983/http://aolsearch.aol.co.uk/article.adp?query=">

  man_update_display

  var_string_get <A>

  uni_use_last_atom_string <man_append_data>

  man_update_display

  man_append_data <"&aol_first=1&aol_last=10&test=0&channel=ADVHealth">

  man_update_display

  man_end_context

  act_do_action <99>

  uni_sync_skip <1>

  async_alert <warning, "Please enter a search word or phrase">

uni_end_stream

uni_sync_skip <2>

uni_end_stream

Now there are only two new atoms in all of that: 

if_stra_neq_null_then <1, 2>

This atom is similar to the conditional atom we looked at in the last section. This time the atom is checking to see if the variable is not empty.

The next atom is brand new:
act_do_action <99>

This atom belongs to the Action Protocol and it looks for a pre-defined action stream with the Id <99> as an event on the current object and executes it. Next, we need to define that event. Get the properties box of the form and open the ‘action tab’. Click on the ‘event’ menu and type in ‘99’. Next set the action to FDO and paste in the following atom stream: (See fig.9)
uni_start_stream

  buf_start_buffer <token_header | stream_id_header | host_bound | start_stream_header | end_stream_trailer | response_id_header>

  buf_set_token <"ur">

  de_start_extraction

  man_set_context_relative <271>

  de_get_data_pointer

  man_end_context

de_end_extraction

  buf_close_buffer

uni_end_stream

Figure 9
This is a subroutine that initiates a series of ‘buffer commands’. To put it simply, Buffer commands are used to pass information to the host. Let’s go through the new atoms one at a time.

buf_start_buffer <token_header | stream_id_header | host_bound | start_stream_header | end_stream_trailer | response_id_header>

This atom marks the start of a buffer subroutine, Is the Buffer Protocol’s equivalent of ‘de_start_extraction’. The long list of values contained in brackets defines the header of the message that is about to be sent to the host. 

When the host receives the data, this header warns the host about the type of data that it contains and how the host should respond to it, similar to the <0> value after de_start_extraction. 

buf_set_token <"ur">

This atom tells the octopus what part of the host should receive the data, similar to the de_set_data_type <text> atom. It sets the token to “ur”, which identifies it as a URL. If we wanted to pass the data as a Keyword for example we would use the same atom stream but change the token value to “Kk”.  (For more explanation about tokens, see the note at the end of the chapter).

de_start_extraction

man_set_context_relative <271>

de_get_data_pointer

man_end_context
These atoms load the content of <271> into the current register.

The next new atom is;

buf_close_buffer

This atom initiates the data transfer. When this atom is executed the contents of the current register will be sent to the Host according to the pre-set state of the buffer tool, as defined by the previous atoms.

Lets look at that in english:

uni_start_stream
OK Listen up!

buf_start_buffer <token_header | stream_id_header | host_bound | start_stream_header | end_stream_trailer | response_id_header>


In a minute I’m going to ask you send some data to the host, here’s the header I want you to put on it. 

buf_set_token <"ur">


Give it to the part of the host that deals with URLs

de_start_extraction


In a minute I’m going to ask you to extract some data

man_set_context_relative <271>


Look at the invisible input box

de_get_data_pointer


Extract the data and place it in the current register

man_end_context


Go back to the form level

buf_close_buffer


Send the contents of the current register to the host as specified earlier

uni_end_stream
Ok stop.

That’s your basic introduction to using FDO in VPD. The next few chapters will concentrate on specific FDO solutions to specific VPD problems like custom browsers, tickers, dynamically changing the content of list boxes and designing forms in non-visual mode.

Note 1: What the heck is a Token?
Good question. If you’ve been creating forms on the AOL service for a while you have no doubt come across Token / Arg pairs before. Put simply, a Token /Arg pair identify a particular application on the AOL Host and pass information to it. For example, the token “ur” identifies an application on the AOL host that goes to the web, retrieves URLs and passes them to the client to be displayed in a browser window. In this case the Arg part of the Token / Arg pair would be the URL that is to be displayed.

On VPD forms, we usually use buffer atoms to set up a transmission to the host. The buf_set_token atom identifies the Token the data is to be sent to and the buf_close_buffer atom sends the data as the Arg part of the Token/ Arg pair. If you have not specified a particular arg within the buffer routine, the token will be sent with the last returned value as the arg. (Just to confuse things a bit more, not all tokens require an arg).

(That’s a very very basic outline of what a token is and how it is used. More information about tokens and how they can be used is sadly beyond the scope of this document.)

Chapter 5 

The D is for Dynamic

Right, so now you should know the basics. You have learnt the main principles behind using the FDO language to add functionality and manipulate your VPD forms. The rest of this document is aimed at showing you examples of FDO code in action which should arm you with a basic arsenal of ‘Click tricks’ that you can use and hopefully inspire you to go away and create your own.

We’re going to start by showing you ways to make your forms ‘dynamic’ rather than static. 

Click trick 2. The dynamic browser

In the last chapter we built a form that launched an AOL web based channel search and displayed the results in the client’s default browser window. That’s all well and good but unfortunately it means that we are unable to track members that go from here to the web. If we could display these web pages inside a custom browser, we would be able to keep our AOL Health channel branding on the top of the page and assign a plusgroup for the browser form so that our Health Channel could keep its clicks.

The problem here is that because the search URL depends on what the member types as a search string, we don’t know what the URL is going to be before the member types it in. So how can we create a custom browser that points to the right place.

The solution lies in FDO’s global variables. 

Lets recap. We learnt in the last chapter that we can use FDO to extract text from a form and store it as a Variable in the A, B, C or D registers. We also learnt that these variables are Global.  Because a global variable can be recalled from anywhere inside the AOL root, we can extract data from one form, save it as a variable and recall it onto another form.

Here are the steps we will need to make the dynamic browser work:

1. Extract the Search text that the member has typed.

2. Use the search text to make a URL in a hidden field on the form.

3. Extract the complete URL and save it as a string variable.

4. Check to see if the browser is already on screen

5. If not, call up a new custom browser form.

6. Recall the URL from the global string variable and load it into the browser’s web field.

To complete this click trick you will need:

1. A form that launches a channel search, (We’re going to use a copy of a UK Travel form).

2. A Custom browser. (A VPD form with a web field on it).

There are certain FDO routines that should be included on most custom browsers as default to allow for resizing and secure connection options etc. I’m going to assume that you have already created a browser form that contains these elements. They will not interfere with our code so we don’t need to go into them in detail. For more information about creating custom browsers, please refer to the VPD manual or you’re local training department. 

Step 1 – Extract the Search String

This is a simple data extraction routine that extracts the contents of the input box and stores it in a variable.

Here’s the code:

uni_start_stream

  de_start_extraction <0>

  man_set_context_relative <8>

  de_get_data_pointer

  uni_save_result

  var_string_set_from_atom <A>

  if_stra_neq_null_then <1, 2>

  man_end_context

The conditional protocol  atom if_stra_neq_null_then <1, 2>, checks to see if the input box is empty and if it is, it executes an error checking atom contained between the uni_sync_skip <1> and uni_sync_skip <2> atoms.

uni_sync_skip <1>

async_alert <warning, "Please enter a search word or phrase. ">

uni_end_stream

uni_sync_skip <2>

Step 2 Use the search text to make a URL in a hidden field on the form

man_set_context_relative <94> (Hidden field)

  man_replace_data <"http://aolsearch.aol.co.uk/article.adp?query=">

  man_update_display

  var_string_get <A>

  uni_use_last_atom_string <man_append_data>

  man_update_display

  man_append_data <"&aol_first=1&aol_last=10&test=0&channel=ADVTravel">

  man_update_display

  man_end_context

This part of the stream constructs the search URL by replacing and appending the contents of the input field <94> with the three parts of the search URL. If you need to make a URL that requires a different syntax you will need to alter this part of your code to match.

Step 3 - Extract the complete URL and save it as a string variable

de_start_extraction <0>

  man_set_context_relative <94>

  de_get_data_pointer

  var_string_set_from_atom <D>

  man_end_context

The above atoms are another simple data extraction routine which extracts our completed URL and saves it in the string area of the D register. When the browser form opens it will take the contents of this variable and try to load it as a URL. We deliberately chose to use the D registry because it is the least used registry on the UK service so the possibility of encountering a conflict with two forms both using the same variable for different purposes is less likely.

Step 4 - Check to see if the browser is already on screen

man_preset_gid <45-10892>

  if_last_return_true_then <3, 4>

  man_make_focus

  act_do_action <100>

  uni_sync_skip <3>

  act_do_action <101>

  uni_sync_skip <4>

Before we call the browser form, we need to make sure it is not already on screen. This will avoid having multiple copies of the same browser on the screen at the same time. The above code checks to see if the form is on screen and if it is, it calls it to the front and then executes the routine that loads the D string variable into the web-field. We’ve not covered this before so lets go through this line by line.

The most important atom here is man_preset_gid <45-10892>

This atom pre sets the GID of the next object created to the specified GID but if an object with that GID already exists on the screen, it will set context to that object instead. Also, this atom returns as true if the predefined GID is already on screen and false if it is not. Which makes it a very handy atom indeed!

Watch out for ‘man_set_context_globalid’
This atom is not as useful as you might expect. Basically it will set the context to an object with the specified GID if it is already on screen. However if the object is not already on screen it will terminate the stream. So its important not to get man_preset_gid and man_set_context_globalid mixed up.

So, lets have a detailed look at this routine:

man_preset_gid <45-10892>
See above, and return true / false

if_last_return_true_then <3, 4>
If true do everything between here and <3>

If false do everything between <3> and <4>

man_make_focus <45-10892>
Bring 45-10892 object to the front

act_do_action <100>
Do action 102 on the current object (Browser)

uni_sync_skip <3>


act_do_action <101>
Do action 101 on the current object (Search form)

uni_sync_skip <4>


That’s all the code you need on the search button of your form. We’ll look at event 101 on the search form in a minute. First, lets see how this all looks together:

uni_start_stream

  de_start_extraction <0>

  man_set_context_relative <8>

  de_get_data_pointer

  uni_save_result

  var_string_set_from_atom <A>

  if_stra_neq_null_then <1, 2>

  man_end_context

  man_set_context_relative <94>

  man_replace_data <"http://aolsearch.aol.co.uk/article.adp?query=">

  man_update_display

  var_string_get <A>

  uni_use_last_atom_string <man_append_data>

  man_update_display

  man_append_data <"&aol_first=1&aol_last=10&test=0&channel=ADVTravel">

  man_update_display

  man_end_context

  de_start_extraction <0>

  man_set_context_relative <94>

  de_get_data_pointer

  var_string_set_from_atom <D>

  man_end_context

  de_end_extraction

  man_preset_gid <45-10892>

  if_last_return_true_then <3, 4>

  man_make_focus <45-10892>

  act_do_action <100>

  uni_sync_skip <3>

  act_do_action <101>

  uni_sync_skip <4>

uni_end_stream

uni_sync_skip <1>

async_alert <warning, "Please enter a search word or phrase. ">

uni_end_stream

uni_sync_skip <2>

uni_sync_skip <4>

uni_end_stream

Step 5 - Call up a new custom browser form

This is the easiest step of all. 

This will call the browser onto the screen.

Step 6 - Recall the URL from the global string variable and load it into the browser’s web field

This routine is placed on the browser form as event 101.

uni_start_stream

  man_set_context_relative <1>

  var_string_get <D>

  uni_use_last_atom_string <www_load_url>

  man_end_context

  man_update_display

uni_end_stream

Now the only thing that remains is to make the browser execute event 100 as soon as it loads like so:

Now as long as you have a plus group hard-coded onto the browser form, you can keep all your web based searches under your channel branding and keep the clicks!

Chapter 6 

The most wonderful thing about tickers is…

FDO based tickers are used throughout the AOL UK service, usually to provide links to articles or stories provided by third party feeds. In this chapter we’re going to look at some code that allows you to place tickers on your forms, particularly Rainman integrated tickers, and the problems associated with them.

First of all, Let’s look at the ingredients of a ticker.

1. A list box with an ID of <30>

This list box will contain links to the different articles or objects, which the ticker will display. If your ticker is going to be a Rainman version, this list box will need to be mapped appropriately. If your ticker is going to be hard coded, you can go ahead and place your list items in here and link them to their destinations. For our example, we will be building a hard coded ticker first and then discuss what we need to do to make this work in Rainman. On the finished version of the form, this listbox will be invisible to the user but for now, we’re going to 

2. A button with an ID of <150>

This is the button that acts as the display of the ticker. As the code runs through the actions, the title of this button will change accordingly.

3. A wingding with an ID of <117>

A winding is a small button that typically sits beside another button or rich text field and points to the same place. Weather you use Wingdings or not is a matter of style and taste really. For the sake of completion, we’re going to use one here. We’ll set it’s action to the following:

uni_start_stream 

man_set_context_relative <150> 

act_do_action <1> 

man_end_context 

uni_end_stream

Right now that we have all the objects that we need, lets take a step back, think about what we need to do and break it into a series of steps just like we did with the dynamic browser in the last chapter.

The steps of s a ticker:

1. Look at the first object in the list box.

2. Extract it’s values into a variable.

3. Replace the title of Button 150 with the title of the list item.

4. Replace the action of Button 150 with the action of the list item.

5. Wait for a few seconds

6. Look at the next Item in the listbox.

7. Repeat steps 2-6.

8. When there are no more items left to look at in the listbox, go back to Step 1.

Now, it’s worth taking some time to look at the above steps and make sure they make sense. You don’t need to worry about how exactly we get everything to work, just make sure that you understand each step. If you don’t understand the steps, there’s no point in going any further because from here on in it gets more complicated.

Now if we look at the above steps we have a better idea of what we need to get our FDO code to do. First of all it needs to be able to find the first object in a listbox, then it needs to be able to tell when there are no more items left in the new box in order to go back and start at the first one again. Once we have the method to control this in place, its just a matter of repeating the same steps over and over again. So how do we find the first item in a list box? Well its really easy actually thanks to the atom 

mat_value <1>

This atom associates a particular value, in this case '1' with the current object. Not every object can have a value defined for it like this, this atom only applies to objects that contain children. In this case we will be using this atom to change the value of the list box.

Any object that contains children objects, (listboxes, pop-up menus, context menus and groups) will automatically have a default value set when the form is created of <1>. This is why when a list box gains focus on a form, the first item in the list is selected by default. We can change which item is selected as default by changing the value of the listbox using the mat_value atom.

In this example the list box has the relative ID <30> and three popup items within it as children.  

Pop up 1 is the first item in the list so when the form is displayed, this will be the object that is selected by default.
If we added the atom : mat_value <3> to the in stream of the listbox, then the third item would be displayed as the selected item when the form is displayed.
The second bit is a little more tricky because we need to be able to tell when the there are no more items left in the listbox so that we can go back to the start again. We do this with the help of another new atom:

man_get_child_count

This atom returns the number of objects that are children of the current object. In the above example the last return value would be '3'. This means we can count the number of times we execute steps 2-6 and compare it to the number of children in the listbox. When the number of times we have run steps 2-6 is the same as the number of children in the listbox we know that we have run through the entire list and so must start again with the first object. Right, this is beginning to confuse even me so lets start looking at the code we will be using before we start to bleed from our ears.

Our first bit of code goes on event <60> at the form level and it looks like this:

uni_start_stream

  var_number_set <A, 0>

  var_number_save <A, 144>

  man_set_context_relative <30>

  man_get_child_count

  var_number_set_from_atom <B>

  man_end_context

  var_number_save <B, 145>

  if_numb_false_then <10, 11>

  man_set_context_relative <150>

  mat_bool_disabled <yes>

  man_end_context

  man_update_display

  mat_timer_duration <5000>

  mat_timer_event <60>

  man_enable_one_shot_timer

  uni_sync_skip <10>

  man_set_context_relative <150>

  mat_bool_disabled <no>

  act_replace_select_action

    <

    uni_start_stream

    sm_do_list_action <30>

    >

  man_end_context

  man_set_context_relative <117>

  mat_bool_disabled <no>

  mat_bool_hidden <no>

  man_end_context

  act_do_action <61>

  uni_sync_skip <11>

uni_end_stream

var_number_set <A, 0>

We start off by declaring a number variable <A> with the value of zero. This variable is going to be our counter and it will increment each time we run through steps 2-6. It will be this variable that we compare to the number of children in the listbox to see if we need to start at the top of the listbox again.

var_number_save <A, 144>
Once we have the variable set to 0, we save it in a 'Bucket' with the ID of 144. We have not yet changed the context from the form level with means that once this atom has been executed we will have a variable in the A registry of the form called '144' holding the value 0. 

Saving Variables in 'Buckets'.

I know what you are thinking, why did we do that? Couldn't we have just left this variable as <A> without its own ID? Well the answer is yes, however there is a danger in doing this when working in numbers. Because FDO only allows us to use four registries for global variables, and many of the forms in the AOL client (for example the buddy list window) use these registries to handle numbers. There is a good chance if your form is on screen for any length of time, that there will be a conflict between two different pieces of FDO code writing to the same variable. In these cases it is always safer to store your number variables in a safe place where there is no danger of them being overwritten by a client based atom stream.

man_set_context_relative <30>

man_get_child_count

  var_number_set_from_atom <B>

  man_end_context

  var_number_save <B, 145>

Next we get the number of items in the listbox and save that number as a variable in the <B> registry with an ID of '145'. Later on we will compare the values of this variable and the previous variable in order to determine when we should restart the ticker.

if_numb_false_then <10, 11>

  man_set_context_relative <150>

  mat_bool_disabled <yes>

  man_end_context

  man_update_display

*mat_timer_duration <5000>

*mat_timer_event <60>

*man_enable_one_shot_timer

  uni_sync_skip <10>
This section of code checks to make sure that the list box does contain content and if it does not it will disable the ticker button, wait for five seconds and then restart the ticker at the top of the stream. We include this simple bit of error checking because most of the tickers on the service have their listbox fed from Rainman links created by automatic feeds. The last thing we want is for our form to break if anything goes wrong with the live feed.  

*We will come back to the three atoms here that trigger the 'wait five seconds' bit a little later, for now, just give them a quick once-over.

man_set_context_relative <150>

  mat_bool_disabled <no>

  act_replace_select_action

    <

    uni_start_stream

    sm_do_list_action <30>

    >

  man_end_context

  man_set_context_relative <117>

  mat_bool_disabled <no>

  mat_bool_hidden <no>

  man_end_context

The last section makes sure that both the ticker display button and the wind-ding, are enabled. (Remember regardless of the natural state of the display button, the previous stream may have disabled it, if the listbox was empty at any point). This stream also sets the action of the display button to execute the action of the currently selected object of the list box. This means that the button will always execute the link to the page that is currently showing in the display.

act_do_action <61>

  uni_sync_skip <11>

uni_end_stream

These last atoms simply mark the end of this stream and execute a new atom stream which is stored on Event 61 of the form.

Event 61

This atom stream simply performs the comparison between our two variables and either moves on to the last section or restarts the ticker code. This is what it looks like:

Event 61

uni_start_stream

  var_lookup_by_id <A, 144>

  var_lookup_by_id <B, 145>

  if_numa_gte_numb_then <12, 13>

  act_do_action <60>

  uni_sync_skip <12>

  act_do_action <62>

  uni_sync_skip <13>

uni_end_stream

 var_lookup_by_id <A, 144>

This atom returns the contents of a 'bucket' variable and loads it into the appropriate register (In this case register A). In this case it loads the value contained in the variable <A,144> into the A section of the current register, which if you remember is equal to the number of list items we have displayed in the ticker. 

var_lookup_by_id <B, 145>

Likewise this atom loads the contents of <B, 145> (The total number of links in the listbox) into the B section of the current register. These two atoms are the equivalent of :

var_number_set <A, (number of links already displayed)>

var_number_set <B, (number of links in the listbox)>

Our next atom compares the values and sends us in a different direction depending on the result:

if_numa_gte_numb_then <12, 13>

This atom can be translated as : If the number of links already displayed is greater than or equal to the number of links in the listbox, then execute all the atoms between here and uni_sync_skip <12>, if not then execute all the atoms between uni_sync_skip <12> and uni_sync_skip <13>.
The only other atoms worth talking about here are the act_do_action <60> and act_do_action <62> lines, which send us in a different direction depending on the result of the conditional statement above.

So over all we can translate the conditional statement as follows:

This atom can be translated as : If the number of links already displayed is greater than or equal to the number of links in the listbox, then go to Event <60>(Start again) if not then go to Event <62> (Carry on).

Next lets look at Event <62> our last Atom stream, which does all the work.

Event 62

uni_start_stream

  var_lookup_by_id <A, 144>

  man_set_context_relative <30>

  var_number_increment_save

  uni_use_last_atom_value <mat_value>

  man_update_display

  de_start_extraction <0>

  de_get_data_pointer

  var_string_set_from_atom <C>

  man_end_context

  var_number_save <A, 144>

  de_end_extraction

  man_set_context_relative <150>

  var_string_get <C>

  uni_use_last_atom_string <man_replace_data>

  man_update_display

  man_end_context

  mat_timer_duration <3500>

  mat_timer_event <61>

  man_enable_one_shot_timer

uni_end_stream

We start off by calling back into memory the value of <A,144>, then we set context to the listbox.

Next we advance our counter by one using:

var_number_increment_save

This atom increases the value of the current variable by one and saves it again, so at this we will have a variable <A> and a return value which is equal to the number of links we have displayed so far plus one. 

The next atom: uni_use_last_atom_value is the numeric equivalent of the 'uni_use_last_atom_string' atom. It executes another atom (which is used as an argument) with the last returned value.

This will have the same effect as executing:

 mat_value < The number of links we have displayed so far plus one >

Which changes the value of the listbox so that the next item in the list is the selected item. If someone clicked on our wingding or our ticker at this point then this item is the one that would be executed so now we need to change the display to reflect this.

de_start_extraction <0>

  de_get_data_pointer

  var_string_set_from_atom <C>

  man_end_context

  var_number_save <A, 144>

  de_end_extraction

This is a simple data extraction routine that extracts the text from the current object (the currently selected object in the listbox), saves it as a string variable <C> and then saves the new value of A back into it's bucket.

man_set_context_relative <150>

  var_string_get <C>

  uni_use_last_atom_string <man_replace_data>

  man_update_display

  man_end_context

This bit of code sets context to our display button and simply uses the contents of the <C> variable to change the text that it is displaying.

We're almost there now. We've managed to find the right object in the list box, change the value  of the selected object to that object and change the display to reflect this. Our next step is to wait for a few seconds, to give the viewer a chance to see the new option in the ticker and then move on to the next object. To do this we need to use a timer routine.

There are three parts to any FDO timer.

1. The length of time.

(I'm going to ask you to count to X)

2. The associated action.
(When you're done counting, do this)

3. The trigger.


(Start counting)

In our case we want to wait for 3.5 seconds, then execute event <61> which will check the values of our variables and either move onto the next item in the listbox or start again. 

And these are the atoms we use to do that:

mat_timer_duration <3500>
(Time is expressed in milliseconds)

mat_timer_event <61>

man_enable_one_shot_timer

Now I know what you're thinking, 'Hang on I've seen those atoms before'. And you have indeed. Remember Event <60>?

*mat_timer_duration <5000>

*mat_timer_event <60>

*man_enable_one_shot_timer

These atoms were used to set up a timer which causes the FDO to keep checking the contents of the listbox in case it is ever empty for any reason. This timer waits for 5 seconds and starts the whole process again so that if a new link suddenly appears in the list box it will automatically start the ticker working again. This is usually only necessary when dealing with Tickers on Rainman canned forms with listboxes that are populated by external feeds etc.

The Last Bit…

So now we should have a perfectly working ticker, but if you've followed these steps and you've previewed your form you will see that nothing actually happens. That’s because we have a series of FDO streams that work the ticker, stored as events on the form level but we have not added any code to initiate the streams yet. To do this we can simply add the following atom to the Create event of the form:

act_do_action <60>

This will begin the ticker code right form the moment the form is displayed, but for a Rainman canned form, this is not really what we want because of the slight delay between the form appearing and the Rainman content appearing in the list box. So to avoid the apparent error of an empty list box, we simply add another timer to the create event before we execute event <60>, like so:

mat_timer_duration <3500>

mat_timer_event <60>

man_enable_one_shot_timer 

This means that the ticker won't start until the form has been on screen for 3.5 seconds, giving our Rainman content plenty of time to appear in the list box.

Now if you've followed and understood this chapter you should be capable of creating a brand new ticker from scratch, and that’s OK as far as it goes. Now who wants to delve a little deeper and fine-tune their ticker with some extra error checking code and making it, dare I say it - dynamic?

Here's a little something I prepared earlier…..

The wonderful thing about tickers, as we have discovered already, is that once they have been built correctly, you can sit back and watch your CGI feeds do all the work. The down side is that you are usually relying on someone else's Feed to keep your ticker ticking. Sometimes feeds break, sometimes through no fault of your own, your ticker doesn't have any content to tick through. Sure, we included some extra code to disable the button if the list box is empty but I never really liked that approach. It’s a bit messy. (Also even though you and I know that your ticker is working as designed, you'll still get the call from the producer who wants to know why her ticker isn't working and what you can do to fix it). Well here's a little trick to avoid these nasty little surprises. For want of a better description, we're going to call it a Dynamic Ticker.

First lets go over an example of how and why you might want to use this approach to ticker building.

UK Election Night News Ticker  - A dynamic ticker at work.

We're expecting a general election sometime later this year in the UK and one of the forms we want to have on the night is a simple ticker that displays the results of each constituency as they are declared. The plan is to have a feed that generates a link in the ticker to the page for that constituency and updates that page. 

Sounds simple enough, however there are some complications.

Problem No1- Too many links

There are over 650 constituencies in the UK, which means that if we add a new link to the listbox for each result, there will be over 650 items linked onto that page by the time all the results have been declared. Now for a start that’s not possible because there is a limit to the amount of items you can have in a Rainman listbox. (300) Also it wouldn't make for a very nice ticker, you would have to wait for hundreds of links to tick by before you see the one that interests you.

Answer No 1 - Remove links after a specified length of time.

The simple answer to this is to limit the length of time that each link appears in the listbox. That’s easy to do using the \remove command in Rainman when the link is created. We can specify how long we want a particular link to be visible for and then make sure that it disappears from the Ticker when it has become 'old news'. 

Problem No 2 - Too few Links

The second problem we face has to do with the time that each result will be posted. Its more than likely that the results will be coming in fits and starts. At the beginning of the night, there will be no results in and the Ticker will therefore be empty. Later on we might get a flurry of results in a small space of time or there might be a dirth of stories in the early hours of the morning. This means that its entirely possible that from time to time the Ticker's listbox will be empty. Which on a traditional Ticker wouldn't be a big problem as if its been built correctly, the ticker will remain disabled until there is something to display, but our form is just a ticker. There will be no other content on the page, so it’s a bit pointless having a form that displays a disabled ticker that doesn't tick and nothing else.

Answer No 2 - Provide default links if the listbox is empty

What we want to do is have some alternate links that the ticker can display if the listbox is empty. The easiest way of doing that is to have two separate listboxes on the form. The first listbox. Lets call it listbox A, is the one that will contain the Rainman links that are created by our live feed. The second listbox (listbox B) will contain hard coded links to default news stories. These stories can be edited and updated through Rainman manually if needed and they will only be displayed in the ticker if there are no recent updates.

Problem No 3 - Too Many Links again

Given the unpredictable way in which the results links will be created, there is a possibility, if we get too many results in a short period of time, that the listbox will contain a large number of links and this will mean it takes to long to tick from start to finish. What if the member misses a link,? he has to wait until the tick comes full circle before he can click it again. If there are say 20 links in the listbox at the time, and the ticker displays each link for 5 seconds, he will have to wait for nearly 2 minutes before he can read the story.

Answer No 3 - Only show the top 10 links

What we need to do is limit the number of items the ticker displays, no matter how many items are in the listbox. In a traditional ticker we use the man_get_child_count atom to count the number of items in the listbox and then use this number to control when the ticker should restart. What we are going to do is set this at a specific value rather than using the number of children. (We've decided to display no more than 10 links in the ticker). If we make sure that the links are displayed with the most recent link at the top, we can be sure that the links displayed will be the ten most recent results.

So that’s all the problems ironed out on paper. Lets look at how we code it.

Event <60>
This is the event that sets up our variables and starts the counter at zero.

uni_start_stream

  var_number_set <A, 0>

  var_number_save <A, 144>

  man_set_context_relative <30>

  man_get_child_count

  var_number_set_from_atom <B>

  man_end_context

  var_number_save <B, 145>

  if_numb_false_then <10, 1>

  var_number_set <D, 31>

  act_do_action <61>

  uni_sync_skip <10>

  var_number_set <D, 30>

  act_do_action <61>

  uni_sync_skip <11>

uni_end_stream

The only difference between this atom stream and the code for Event <60> we looked at in the previous example is the var_number_set <D, 30> which is executed if the list box is not empty. This sets a D variable with the value "30", (which as it happens is the RID of Listbox A). If the listbox is empty, we set the value of D to "31", the RID of listbox B. We're going to use that variable later to direct the ticker to the appropriate listbox.

Next lets look at Event <61> 

Event <61>

This is the event that checks which item the ticker is displaying and either redirects it to the top of the list or moves it along to the next item.

uni_start_stream

  var_lookup_by_id <A, 144>

  var_number_set <B, 10>

  if_numa_gte_numb_then <12, 13>

  act_do_action <60>

  uni_sync_skip <12>

  var_number_get <D>

  uni_use_last_atom_value <man_set_context_relative>
  man_get_child_count

  var_number_set_from_atom <B>

  man_end_context

  ;IF NUMBER OF ITEMS IS LESS THAN 10

  ;RESTART

  if_numa_gte_numb_then <14, 15>

  act_do_action <60>

  uni_sync_skip <14>

  ;ELSE CARRY ON

  act_do_action <62>

  uni_sync_skip <13>

  uni_sync_skip <15>

uni_end_stream

The first atom to draw your attention to here is var_number_set <B, 10> which sets B to the maximum values of stories in the listbox. If the value of A is greater than 10, the ticker will restart rather than move on to the eleventh item.

Next comes the interesting bit:

var_number_get <D>

 uni_use_last_atom_value <man_set_context_relative>
Here we are using the D variable we set earlier to set the context to the appropriate listbox. Once we've done that we reset the value of B to the child count to make sure that the listbox restarts if it reaches the end of the list before it reaches ten. This stops our ticker from breaking if there are less than 10 items in the listbox.

Event <62>

Event <62> is the event that performs all the data extraction and makes the ticker work. The only difference between this and the earlier example is the way in which the stream sets context to the list box using the value of variable D.

uni_start_stream

  var_lookup_by_id <A, 144>

  var_number_increment_save

  uni_save_result

  var_number_save <A, 144>

  var_number_get <D>

  uni_use_last_atom_value <man_set_context_relative>
  uni_get_result

  uni_use_last_atom_value <mat_value>

  man_update_display

  de_start_extraction <0>

  de_get_data_pointer

  var_string_set_from_atom <C>

  man_end_context

  de_end_extraction

  man_set_context_relative <150>

  var_string_get <C>

  uni_use_last_atom_string <man_replace_data>

  man_end_context

  man_update_display

  mat_timer_duration <3500>

  mat_timer_event <61>

  man_enable_one_shot_timer

uni_end_stream

So, that’s Dynamic Tickers for you. You shouldn't have to use dynamic tickers that often. Most of the time its ok to just let your ticker lie sleeping while it waits for something interesting to display. In this case that wasn't acceptable because we were building a form that had no other content. You might think of other example of why you might want to take this approach. The important thing for now is that you know you can if you want to.

Chapter 7 

How long is a piece of a string?

So far we have managed to extract strings of characters and numbers and manipulated them in various ways but we have only ever used them in their whole state. Wouldn't it be useful if we could split those strings into smaller strings? Or wouldn't it be nice to be able to use FDO to count the number of characters in a string? In fact, once you get into it, there's lots of things you may want to do to a string once you have extracted it from a form. In this chapter we're going to talk a bit about how FDO handles strings and how we can take advantage of this to develop all manner of click tricks. 

It's not a string, it’s a number.

Computers as we all know are dumb. They don't understand words, characters or letters. The only thing a computer understands, (and by extension, our octopus) is numbers. That is a huge drawback but its made slightly easier to deal with by the fact that its very very good with numbers. So its important to understand that when we ask the octopus to go get a string from an input field, in reality its really just getting a number. This number is called a pointer.

When the user types a string onto a form, the computer stores those characters as they are typed into its local short-term memory. In visual terms it might look something like this:

The pointer tells the octopus where to look for the string. In the above example the string "String" is stored at the memory address that begins at byte 179265453. So in this case, our pointer has the value "179265453". In the computer's memory, the end of a string is marked with a zero, so the octopus goes to the right memory location and reads the data until it finds a zero and then stops. This is what is known as a Null-Terminated String. 

Lets look at how this might work in the middle if an atom stream. Imagine that we have a form on screen that has an input field (RID <10>) which contains the characters "string".

…

man_set_context_relative <10>

de_start_extraction <0>

de_get_data_pointer

var_string_set_from_atom <A>

de_end_extraction

Now you should already know what this will do, but here's what's really happening…

man_set_context_relative <10>

(Return value = true)

de_start_extraction <0>


(Return value = unchanged)

de_get_data_pointer


(Return value = 179265455)

var_string_set_from_atom <A>

(Sets the value of <A> as "Contents of 179265455") 

de_end_extraction


(Return value = unchanged)

Get it? The de_get_data_pointer does not return the string itself, it returns a number that tells the octopus where to find the string.

Now, knowing this we can be a bit clever. Try this on for size:

de_start_extraction <0>

de_get_data_pointer

var_number_set_from_atom <A>

de_end_extraction

You see what we have done here? We've extracted the pointer and saved it as a number variable and of course once we have a number variable, we can manipulate it further like this:

man_set_context_relative <10>


(Return value = true)

de_start_extraction <0>



(Return value = unchanged)

de_get_data_pointer



(Return value = 179265455)

var_string_set_from_atom <A>

(Sets the value of <A> as "Contents of 179265455") 

de_end_extraction



(Return value = unchanged)

var_number_increment <A>


(Return value = 179265456)

var_number_increment <A>


(Return value = 179265457)

man_set_context_relative <11>


(Return value = true)

var_number_get <A>



(Return value = 179265457)

uni_use_last_atom_string <man_replace_data>  (replaces the contents of <11> with "ring")

Likewise we can increment and decrement the pointer to show whichever part of the string that we wish. If we only want to use the first few characters of a string, then all we have to do is dump it into an input box, which has its capacity, limited to the number of characters that we want. So if field <11> has its capacity limited to only accept two characters, then the result of the above stream would only display the letters "ri". We can then extract the contents of field <11> and create a variable, which only contains the middle two characters of the word "string".

How long is a piece of string?

Armed with this knowledge we can go a little further and use pointers to determine the length of a string. Remember, the end of a string is marked with a zero, so all we have to do is keep incrementing the pointer and setting a variable until that variable is equal to zero (null). If we increment another variable that counts how many times we have done this, that variable will have the same value as the number of characters.  The easiest way to do this is by setting up a loop. Here's the stream:

uni_start_stream

man_set_context_relative <10>

  de_start_extraction <0>

  de_get_data_pointer

  var_number_set_from_atom <A>

  man_change_context_relative <11>

  var_number_set <B, 0>

  uni_start_loop

  var_number_get <A>

  var_number_increment <A>

  var_string_set_from_atom <A>

  if_stra_neq_null_then <1>

  var_number_increment <B>

  uni_end_loop

  uni_sync_skip <1>

  man_set_context_relative <11>

  var_number_get <B>

  uni_convert_last_atom_data

  uni_use_last_atom_string <man_replace_data>

  man_end_context

  man_update_display

uni_end_stream

Yea yea, I know what your asking now, "What the heck is a loop?"

Loops and how to use them

The uni_start_loop and uni_end_loop atoms, allow you to specify a sub routine of atoms that will be repeated until a particular condition is met. Here's a more basic example:

uni_start_stream

  var_number_set <A, 10>

  var_number_set <B, 0>

  uni_start_loop




Marks the start of the atoms to be repeated

  man_set_context_relative <1>

  var_number_get <B>

  uni_convert_last_atom_data

  uni_use_last_atom_string <man_append_data>

  man_append_data <",">

  man_update_display

  var_number_increment <B>

  if_numa_gte_numb_then <1>


Loop is repeated as long as this condition is true

  uni_end_loop




Marks the end of the atoms to be repeated

  uni_sync_skip <1>

  async_alert <warning, "A > B">

uni_end_stream

